(eval-when (:compile-toplevel :load-toplevel :execute)
  (sb-int:defconstant-eqx OPT
    #+swank '(optimize (speed 3) (safety 2))
    #-swank '(optimize (speed 3) (safety 0) (debug 0))
    #'equal)
  #+swank (ql:quickload '(:cl-debug-print :fiveam) :silent t)
  #-swank (set-dispatch-macro-character
           ;; enclose the form with VALUES to avoid being captured by LOOP macro
           #\# #\> (lambda (s c p) (declare (ignore c p)) `(values ,(read s nil nil t)))))
#+swank (cl-syntax:use-syntax cl-debug-print:debug-print-syntax)
#-swank (disable-debugger) ; for CS Academy

;; BEGIN_INSERTED_CONTENTS


(declaim (ftype (function * (values fixnum &optional)) read-fixnum))
(defun read-fixnum (&optional (in *standard-input*))
  "NOTE: cannot read -2^62"
  (macrolet ((%read-byte ()
               `(the (unsigned-byte 8)
                     #+swank (char-code (read-char in nil #\Nul))
                     #-swank (sb-impl::ansi-stream-read-byte in nil #.(char-code #\Nul) nil))))
    (let* ((minus nil)
           (result (loop (let ((byte (%read-byte)))
                           (cond ((<= 48 byte 57)
                                  (return (- byte 48)))
                                 ((zerop byte) ; #\Nul
                                  (error "Read EOF or #\Nul."))
                                 ((= byte #.(char-code #\-))
                                  (setf minus t)))))))
      (declare ((integer 0 #.most-positive-fixnum) result))
      (loop
        (let* ((byte (%read-byte)))
          (if (<= 48 byte 57)
              (setq result (+ (- byte 48)
                              (* 10 (the (integer 0 #.(floor most-positive-fixnum 10))
                                         result))))
              (return (if minus (- result) result))))))))

(defmacro dbg (&rest forms)
  #+swank
  (if (= (length forms) 1)
      `(format *error-output* "~A => ~A~%" ',(car forms) ,(car forms))
      `(format *error-output* "~A => ~A~%" ',forms `(,,@forms)))
  #-swank (declare (ignore forms)))

(defmacro define-int-types (&rest bits)
  `(progn
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "UINT~A" b)) () '(unsigned-byte ,b))) bits)
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "INT~A" b)) () '(signed-byte ,b))) bits)))
(define-int-types 2 4 7 8 15 16 31 32 62 63 64)

(declaim (inline println))
(defun println (obj &optional (stream *standard-output*))
  (let ((*read-default-float-format* 'double-float))
    (prog1 (princ obj stream) (terpri stream))))

(defconstant +mod+ 1000000007)

;;;
;;; Body
;;;

(defun bisect-right (target value pivot)
  (declare #.OPT
           (int32 pivot)
           (fixnum value)
           ((simple-array int32 (*)) target))
  (macrolet
      ((frob (accessor &optional declaration)
         `(progn
            (labels
                ((%bisect-right (left ok)
                   ;; TARGET[OK] > VALUE always holds (assuming
                   ;; TARGET[END] = +infinity)
                   ,@(when declaration (list declaration))
                   (let ((mid (ash (+ left ok) -1)))
                     (if (= mid left)
                         (if (< value (the fixnum (* pivot (,accessor target left))))
                             left
                             ok)
                         (if (< value (the fixnum (* pivot (,accessor target mid))))
                             (%bisect-right left mid)
                             (%bisect-right mid ok))))))
              
              (%bisect-right 0 (length target))))))
    (etypecase target
      (vector
       (frob aref (declare ((integer 0 #.most-positive-fixnum) left ok)))))))

(defun sort< (as)
  (declare #.OPT
           (inline sort)
           ((simple-array int32 (*)) as))
  (sort as #'<))

(defun sort> (as)
  (declare #.OPT
           (inline sort)
           ((simple-array int32 (*)) as))
  (sort as #'>))

(defun main ()
  (declare #.OPT)
  (let* ((n (read))
         (k (read))
         (as (make-array n :element-type 'int32))
         (as+ (make-array n :fill-pointer 0 :element-type 'int32))
         (as- (make-array n :fill-pointer 0 :element-type 'int32)))
    (dotimes (i n)
      (let ((a (read-fixnum)))
        (setf (aref as i) a)
        (if (>= a 0)
            (vector-push a as+)
            (vector-push a as-))))
    (let ((as+< (sort< (coerce as+ '(simple-array int32 (*)))))
          (as-< (sort< (coerce as- '(simple-array int32 (*)))))
          (as+> (sort> (coerce as+ '(simple-array int32 (*)))))
          (as-> (sort> (coerce as- '(simple-array int32 (*))))))
      (declare ((simple-array int32 (*)) as+< as-< as+> as->))
      (labels ((calc (pivot threshold)
                 (declare (int32 pivot)
                          (fixnum threshold))
                 (let ((res 0))
                   (declare (uint62 res))
                   (cond ((and (>= pivot 0) (>= threshold 0))
                          (incf res (the uint31 (bisect-right as+< threshold pivot)))
                          (incf res (length as->)))
                         ((and (>= pivot 0) (< threshold 0))
                          (incf res (the uint31 (bisect-right as-< threshold pivot))))
                         ((and (< pivot 0) (>= threshold 0))
                          (incf res (the uint31 (bisect-right as-> threshold pivot)))
                          (incf res (length as+>)))
                         ((and (< pivot 0) (< threshold 0))
                          (incf res (the uint31 (bisect-right as+> threshold pivot)))))
                   res))
               (calc-count (threshold)
                 (declare (fixnum threshold))
                 (let ((res 0))
                   (declare (uint62 res))
                   (sb-int:dovector (a as)
                     (incf res (calc a threshold)))
                   (sb-int:dovector (a as)
                     (when (< (* a a) threshold)
                       (decf res)))
                   (setq res (ash res -1))
                   res)))
        (sb-int:named-let bisect ((ng most-negative-fixnum) (ok most-positive-fixnum))
          (if (<= (- ok ng) 1)
              (println ok)
              (let ((mid (floor (+ ng ok) 2)))
                (if (>= (calc-count mid) k)
                    (bisect ng mid)
                    (bisect mid ok)))))
        ))))

#-swank (main)

;;;
;;; Test and benchmark
;;;

#+swank
(defun io-equal (in-string out-string &key (function #'main) (test #'equal))
  "Passes IN-STRING to *STANDARD-INPUT*, executes FUNCTION, and returns true if
the string output to *STANDARD-OUTPUT* is equal to OUT-STRING."
  (labels ((ensure-last-lf (s)
             (if (eql (uiop:last-char s) #\Linefeed)
                 s
                 (uiop:strcat s uiop:+lf+))))
    (funcall test
             (ensure-last-lf out-string)
             (with-output-to-string (out)
               (let ((*standard-output* out))
                 (with-input-from-string (*standard-input* (ensure-last-lf in-string))
                   (funcall function)))))))

#+swank
(defun get-clipbrd ()
  (with-output-to-string (out)
    (run-program "powershell.exe" '("-Command" "Get-Clipboard") :output out :search t)))

#+swank (defparameter *this-pathname* (uiop:current-lisp-file-pathname))
#+swank (defparameter *dat-pathname* (uiop:merge-pathnames* "test.dat" *this-pathname*))

#+swank
(defun run (&optional thing (out *standard-output*))
  "THING := null | string | symbol | pathname

null: run #'MAIN using the text on clipboard as input.
string: run #'MAIN using the string as input.
symbol: alias of FIVEAM:RUN!.
pathname: run #'MAIN using the text file as input."
  (let ((*standard-output* out))
    (etypecase thing
      (null
       (with-input-from-string (*standard-input* (delete #\Return (get-clipbrd)))
         (main)))
      (string
       (with-input-from-string (*standard-input* (delete #\Return thing))
         (main)))
      (symbol (5am:run! thing))
      (pathname
       (with-open-file (*standard-input* thing)
         (main))))))

#+swank
(defun gen-dat ()
  (uiop:with-output-file (out *dat-pathname* :if-exists :supersede)
    (format out "")))

#+swank
(defun bench (&optional (out (make-broadcast-stream)))
  (time (run *dat-pathname* out)))

;; To run: (5am:run! :sample)
#+swank
(it.bese.fiveam:test :sample
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "4 3
3 3 -4 -2
"
    "-6
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "10 40
5 4 3 2 -1 0 0 0 0 0
"
    "6
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "30 413
-170202098 -268409015 537203564 983211703 21608710 -443999067 -937727165 -97596546 -372334013 398994917 -972141167 798607104 -949068442 -959948616 37909651 0 886627544 -20098238 0 -948955241 0 -214720580 277222296 -18897162 834475626 0 -425610555 110117526 663621752 0
"
    "448283280358331064
")))


Measured memory_bytes: 44008